

## Exercice 1 : Profondeur d'une étiquette

Dans cet exercice, on considère les arbres binaires, définis en OCAML au moyen du type ci-dessous.

```
1 | type 'a btree =
2 |   | V
3 |   | N of 'a * 'a btree * 'a btree
```

**Q. 1** Définir une fonction `profondeur_etiquette : 'a btree -> 'a -> int option` prenant en paramètres un arbre binaire et une étiquette  $x$  et retournant la profondeur du nœud le plus profond contenant l'étiquette  $x$ . Si un tel nœud n'existe pas, on retournera `None`.

### Solution

```
1 | let rec profondeur_etiquette (a: 'a btree) (x: 'a): int option =
2 |   match a with
3 |   | V -> None
4 |   | N(y, g, d) ->
5 |     match profondeur_etiquette g x, profondeur_etiquette d x with
6 |     | Some(hg), Some(hd) -> Some ((max hg hd) + 1)
7 |     | None, Some(h) | Some(h), None -> Some h
8 |     | None, None -> if y = x then Some 0 else None
```

On considère à présent les arbres généraux, définis ci-dessous.

```
1 | type 'a gtree = GN of 'a * 'a gtree list
```

**Q. 2** Définir une fonction `profondeur_etiquette : 'a gtree -> 'a -> int option` prenant en paramètres un arbre général et une étiquette  $x$  et retournant la profondeur du nœud le plus profond contenant l'étiquette  $x$ . Si un tel nœud n'existe pas, on retournera `None`.

### Solution

```
1 | let rec profondeur_etiquette (a: 'a gtree) (x: 'a): int option =
2 |   let GN(y, l) = a in
3 |   match profondeur_etiquette_liste l x with
4 |   | Some(h) -> Some(h)
5 |   | _ -> if y = x then Some(0) else None
6 | and profondeur_etiquette_liste (al: 'a gtree list) (x: 'a): int option =
7 |   match al with
8 |   | [] -> None
9 |   | a :: al' ->
10 |     match (profondeur_etiquette a x), (profondeur_etiquette_liste al' x) with
11 |     | Some(hg), Some(hd) -> Some ((max (1 + hg) hd))
12 |     | None, Some(h) -> Some(h)
13 |     | Some(h), None -> Some (h+1)
14 |     | None, None -> None
```

## Exercice 2 : Sous-arbres

Dans cet exercice, on considère les arbres binaires, définis en OCAML au moyen du type ci-dessous.

```
1 | type 'a btree =  
2 |   | V  
3 |   | N of 'a * 'a btree * 'a btree
```

Q. 1 Définir une fonction `egaux : 'a btree -> 'a btree -> bool` prenant en paramètres deux arbres et testant s'ils sont égaux.

Solution

```
1 | let egaux (a: 'a btree) (b: 'a btree) = a = b
```

On dit d'un arbre  $a$  que c'est un sous-arbre d'un arbre  $b$  dès lors que l'arbre  $a$  apparaît dans l'arbre  $b$ . Formellement, on peut définir l'ensemble des sous-arbres inductivement de la manière suivante :

$$\begin{aligned} \text{sous\_arbres}(E) &= \{E\} \\ \text{sous\_arbres}(N(x, g, d)) &= \{N(x, g, d)\} \cup \{\text{sous\_arbres}(g)\} \cup \{\text{sous\_arbres}(d)\}. \end{aligned}$$

Q. 2 Définir une fonction OCAML `sous_arbres : 'a btree -> 'a btree -> bool` permettant de tester si un arbre  $a$  est un sous-arbre d'un arbre  $b$ .

Solution

```
1 | let rec sous_arbres (a: 'a btree) (b: 'a btree) =  
2 |   (egaux a b) ||  
3 |   (match b with  
4 |     | V           -> false  
5 |     | N(x, g, d) -> (sous_arbres a g) || (sous_arbres a d))
```

On considère à présent les arbres généraux, définis ci-dessous.

```
1 | type 'a gtree = N of 'a * 'a gtree list
```

Q. 3 Définir une fonction OCAML `sous_arbres : 'a gtree -> 'a gtree -> bool` permettant de tester si un arbre général  $a$  est un sous-arbre d'un arbre général  $b$ . *Indication* : On rappelle l'existence de la fonctionnelle `List.exists : ('a -> bool) -> 'a list -> bool` prenant en paramètres un prédicat  $p$  et une liste  $l$  et testant s'il existe, dans  $l$ , un élément  $x$  tel que  $p(x)$  est vrai.

Solution

```
1 | let rec sous_arbres (a: 'a gtree) (b: 'a gtree) =  
2 |   (a = b) ||  
3 |   (let N(_, l) = b in  
4 |     List.exists (fun x -> sous_arbres a x) l)
```

## Exercice 3 : Meilleure branche

Dans cet exercice, on considère les arbres binaires, définis en OCAML au moyen du type ci-dessous.

```
1 | type 'a btree =  
2 |   V  
3 |   N of 'a * 'a btree * 'a btree
```


Une *branche* dans un arbre binaire est un chemin entre la racine et une feuille de l'arbre. On définit la *valeur* d'une branche  $(n_1, n_2, \dots, n_p)$  comme la somme des étiquettes des nœuds des branches.

**Q. 1** Définir une fonction OCAML `val_pg_branche: int btree -> int` calculant la valeur de la plus grande branche d'un arbre binaire étiqueté par des entiers.

Solution

```
1 | let rec val_pg_branche (b: int btree): int =  
2 |   match b with  
3 |   | V           -> 0  
4 |   | N(x, g, d) -> x + max (val_pg_branche g) (val_pg_branche d)
```

On représente une branche, dans un arbre binaire, en OCAML, au moyen d'une liste de booléens, comme indiqué dans la figure ci-dessous.



**Q. 2** Définir une fonction OCAML `pg_branche: int btree -> (int * (bool list))` calculant à la fois la valeur d'une plus grande branche et une plus grande branche d'un arbre binaire étiqueté par des entiers.

Solution

```
1 | let rec pg_branche (b: int btree): int * (bool list) =  
2 |   match b with  
3 |   | V -> (0, [])  
4 |   | N(x, g, d) ->  
5 |     let vg, bg = pg_branche g in  
6 |     let vd, bd = pg_branche d in  
7 |     if vg > vd then (x + vg, false :: bg)  
8 |     else           (x + vd, true  :: bd)
```

On considère à présent les arbres généraux, définis ci-dessous.

```
1 | type 'a gtree = GN of 'a * 'a gtree list
```

On représente une branche, dans un arbre binaire, en OCAML, au moyen d'une liste d'entiers : le premier fils est indiqué par 0, le second par 1, le troisième par 2, ....

**Q. 3** Définir une fonction OCAML `pg_branche_g: int gtree -> (int * (intr list))` calculant à la fois la valeur d'une plus grande branche et une plus grande branche d'un arbre général étiqueté par des entiers.

## Solution

```
1 let rec pg_branche_g (g: int gtree): int * (int list) =
2   match g with
3   | GN(x, []) -> (x, [])
4   | GN(x, gs) ->
5     let (v, b), idx = pg_branche_g_aux gs in
6     (v + x, idx :: b)
7
8   (* Parmi la liste d'arbres généraux gs, retourne la valeur de la branche de
9     celui ayant la plus grande branche, et la branche, ainsi que l'indice,
10    dans la liste, de l'arbre dont elle provient. *)
11 and pg_branche_g_aux (gs: int gtree list): (int * (int list)) * int =
12   match gs with
13   | [] -> failwith "[pg_branche_g_aux] liste vide"
14   | g :: [] -> (pg_branche_g g, 0)
15   | g :: gs' ->
16     let v, b = pg_branche_g g in
17     let (v', b'), idx = pg_branche_g_aux gs' in
18     if v' > v then (v', b'), idx + 1
19     else (v, b), 0
```